



На головну ► Мої курси ► Механіко-математичний факультет ► СО ► Тема 8 ►
Лекція 8. Побудова кривої за її кривиною і скрутом. Обчислення вищих кривин кривої в E^n

Лекція 8. Побудова кривої за її кривиною і скрутом. Обчислення вищих кривин кривої в E^n

Тригранник або **репер Френе** — це природна рухома система в тривимірному просторі, що виникає на C^3 -гладкій кривій.

Нехай $\gamma(t)$ — C^3 -гладка крива в Евклідовому просторі \mathbb{E}^3 . Крива задана радіус-вектором $r = r(s)$, де s — натуральний параметр. З точкою ненульової кривини $p \in \gamma(t)$, $p = r(s)$ можна зв'язати три вектори $\tau(s)$, $\nu(s)$, $\beta(s)$, які утворюють ортонормований базис. Де

$\tau = \dot{r}(s)$ — одиничний дотичний вектор,

$\nu = \frac{\ddot{r}(s)}{\|\ddot{r}(s)\|}$ — одиничний вектор головної нормалі,

$\beta = [\tau, \nu]$ — одиничний вектор бінормалі до кривої в даній точці.

Вектори τ, ν, β зв'язані співвідношеннями:

$$\frac{d\tau}{ds} = k\nu$$

$$\frac{d\nu}{ds} = -k\tau + \kappa\beta$$

$$\frac{d\beta}{ds} = -\kappa\nu$$

Величини

$$k = \|\ddot{\gamma}(s)\|, \quad \kappa = -\langle \dot{\beta}, \nu \rangle$$

називають, відповідно, кривиною та скрутом кривої в даній точці. Рівняння виду $k = f(s) \in C^1$, $\kappa = g(s) \in C^0$, де $f(s)$ усюди додатня називаються натуральними рівняннями кривої та визначають її з точністю до руху у просторі. Це твердження називають основною теоремою теорії кривих.

Формули Френе також відомі як *теорема Френе*, можна сформулювати, більш стисло, використовуючи матричні позначення:

$$\begin{bmatrix} \mathbf{T}' \\ \mathbf{N}' \\ \mathbf{B}' \end{bmatrix} = \begin{bmatrix} 0 & \kappa & 0 \\ -\kappa & 0 & \tau \\ 0 & -\tau & 0 \end{bmatrix} \begin{bmatrix} \mathbf{T} \\ \mathbf{N} \\ \mathbf{B} \end{bmatrix}.$$

Ця матриця буде кососиметричною.

З основної теореми теорії кривих випливає, що за кривиною та скрутом можна відтворити криву. Покажемо, як можна в Maple відтворити плоску криву за її кривиною.

Спочатку введемо кутову функцію:

малюнок 8.1

```
>  
> alpha(s)=[Int(cos(theta(u)),u=0..s),Int(sin(theta(u)),u=0..s)];
```

$$\alpha(s) = \left[\int_0^s \cos(\theta(u)) du, \int_0^s \sin(\theta(u)) du \right]$$

Функція $\theta(u)$ є інтегралом від функції $k = k(s)$.

Малюнок 8.2

```
theta(u)=Int(k(t),t=0..u);
```

$$\theta(u) = \int_0^u k(t) dt$$

Приклад 1. Відтворити криву, якщо її натуральне рівняння має вигляд: $k = s$.

Малюнок 8.3


```
sys:=diff(theta(s),s)=s, diff(b1(s),s)=cos(theta(s)), diff(b2(s),s)=sin(theta(s));
```

$$\text{sys} := \frac{d}{ds} \theta(s) = s, \frac{d}{ds} b1(s) = \cos(\theta(s)), \frac{d}{ds} b2(s) = \sin(\theta(s))$$

```
p:=dsolve({sys,theta(0)=0,b1(0)=0,b2(0)=0},{theta(s),b1(s),b2(s)},type=numeric):
```

```
test:=odeplot(p,[b1(s),b2(s)],-5..5,numpoints=200):
```

```
display(test,view=[-2..2,-2..2]);
```



Приклад 2. Відтворити криву, якщо її натуральне рівняння має вигляд: $k = \frac{1}{1+s^2}$.

Малюнок 8.4


```
> sys:=diff(theta(s),s)=1/(1+s^2), diff(b1(s),s)=cos(theta(s)), diff(b2(s),s)=sin(theta(s));
```

$$\text{sys} := \frac{d}{ds} \theta(s) = \frac{1}{1+s^2}, \frac{d}{ds} b1(s) = \cos(\theta(s)), \frac{d}{ds} b2(s) = \sin(\theta(s))$$

```
> p:=dsolve({sys, theta(0)=0, b1(0)=0, b2(0)=0}, {theta(s), b1(s), b2(s)}, type=numeric);
```

```
p := proc(x_rkf45) ... end proc
```

```
> odeplot(p, [b1(s), b2(s)], -8..8, numpoints=200, scaling=constrained);
```



На практиці з диференціальної геометрії зазвичай розв'язують задачі на відтворення плоских кривих. В Maple можна створювати процедури, за допомогою яких відтворюються плоскі та просторові криві з формул Френе.

Приклад 3. Процедура відтворення кривої з формул Френе для плоскої кривої має вигляд:

```
recreate:=proc(kap,a,b,c,d,f,g) local sys,b1,b2,p,theta,pl;
> sys:=diff(theta(s),s)=kap(s), diff(b1(s),s)=cos(theta(s)),
diff(b2(s),s)=sin(theta(s));
> p:=dsolve({sys, theta(0)=0, b1(0)=0, b2(0)=0},
{theta(s), b1(s), b2(s)}, type=numeric):
> pl:=odeplot(p, [b1(s), b2(s)], a..b, numpoints=400, thickness=1, axes=framed,
color=red):
> display(pl, view=[c..d, f..g], scaling=constrained);
> end;
```

```
recreate := proc(kap, a, b, c, d, f, g)
local sys, b1, b2, p, theta, pl;
sys := diff(theta(s), s) = kap(s), diff(b1(s), s) =
cos(theta(s)), diff(b2(s), s) = sin(theta(s));
p := dsolve({b1(0) = 0, b2(0) = 0, theta(0) = 0, sys}, {b1(s), b2(s),
theta(s)}, type = numeric);
pl := plots:-odeplot(p, [b1(s), b2(s)], a .. b, numpoints = 400, thickness =
1, axes = framed,
color = red);
plots:-display(pl, view = [c .. d, f .. g], scaling = constrained) end proc
```


Скористуємось цією процедурою для відтворення кривої з натуральним рівнянням $k = \frac{1}{1+s^2}$

Малюнок 8.5

```
> kap0:=t->1/(1+t^2);
```

$$kap0 := t \rightarrow \frac{1}{1+t^2}$$

```
> recreate(kap0,-10,10,-4,4,0,10);
```



Приклад 4. Відтворити просторову криву за формулами Френе: $k = \frac{1}{5}$, $\kappa = \frac{1}{10}$. Спочатку необхідно створити загальну процедуру в Maple.


```
recreate3dview:=proc(kap,ta,a,b,c,d,e,f,g,h) local
sys,p,alph1,alph2,alph3,T1,T2,T3,N1,N2,N3,B1,B2,B3,pl;
> sys:=diff(alph1(s),s)=T1(s),diff(alph2(s),s)=T2(s),diff(alph3(s),s)=T3(s),
> diff(T1(s),s)=kap(s)*N1(s),diff(T2(s),s)=kap(s)*N2(s),
> diff(T3(s),s)=kap(s)*N3(s),diff(N1(s),s)=-kap(s)*T1(s)+ta(s)*B1(s),
> diff(N2(s),s)=-kap(s)*T2(s)+ta(s)*B2(s),diff(N3(s),s)=-
kap(s)*T3(s)+ta(s)*B3(s),
> diff(B1(s),s)=-ta(s)*N1(s),diff(B2(s),s)=-ta(s)*N2(s),
> diff(B3(s),s)=-ta(s)*N3(s); print(sys);
> p:=dsolve({sys, alph1(0)=0,alph2(0)=0,alph3(0)=0,T1(0)=1,T2(0)=0,T3(0)=0,
> N1(0)=0,N2(0)=1,N3(0)=0,B1(0)=0,B2(0)=0,B3(0)=1},
> {alph1(s),alph2(s),alph3(s),T1(s),T2(s),T3(s),N1(s),N2(s),N3(s),
> B1(s),B2(s),B3(s)},type=numeric):
> pl:=odeplot(p,[alph1(s),alph2(s),alph3(s)],a..b,numpoints=200,thickness=1,
> axes=framed,color=red):
> display(pl,scaling=constrained,view=[c..d,e..f,g..h]);
> end:
```

В команді `recreate3dview`, яка звертається до процедури, вказуємо параметри $kap = \frac{1}{5}$, $ta = \frac{1}{10}$, що відповідають значенню кривини та скруту шуканої кривої

Малюнок 8.6

```
> recreate3dview(1/5,1/10,0,200,0,25,0,20,0,40);
```

$$\frac{d}{ds} \text{alph1}(s) = T1(s), \frac{d}{ds} \text{alph2}(s) = T2(s), \frac{d}{ds} \text{alph3}(s) = T3(s), \frac{d}{ds} T1(s) = \frac{1}{5} N1(s), \frac{d}{ds} T2(s) = \frac{1}{5} N2(s), \frac{d}{ds} T3(s) = \frac{1}{5} N3(s), \frac{d}{ds} N1(s) = -\frac{1}{5} T1(s) + \frac{1}{10} B1(s), \frac{d}{ds} N2(s) = -\frac{1}{5} T2(s) + \frac{1}{10} B2(s),$$
$$\frac{d}{ds} N3(s) = -\frac{1}{5} T3(s) + \frac{1}{10} B3(s), \frac{d}{ds} B1(s) = -\frac{1}{10} N1(s), \frac{d}{ds} B2(s) = -\frac{1}{10} N2(s), \frac{d}{ds} B3(s) = -\frac{1}{10} N3(s)$$



В результаті ми отримали гвинтову лінію. Її кривина та скрут відповідно дорівнюють $\frac{1}{5}$ та $\frac{1}{10}$.

Остання зміна: субота, 20 грудень 2014 22:58

УСТАНОВКИ

Керування модулем сторінок

- Редагувати параметри
- Фільтри
- Події
- Резервна копія
- Відновлення

Керування курсом

Перемикнути на роль...

Мій профіль

